COMPLEX ROOTS & ALL ROOTS OF A POLYNOMIAL

Prof. Richard B. Goldstein

COMPLEX NEWTON-RAPHSON (file: compnrf2.f)

Consider $f(z) = e^z + z^2 = 0$. Clearly there are no real roots since $e^z > 0$ and $z^2 \geq 0$.

By using Newton-Raphson $p_n = p_{n-1} – f(p_{n-1})/f\,'(p_{n-1})$ with complex numbers, including a complex starting point p_0, we will get a sequence of iterates that converge to the root.

Letting $p_0 = 2 + 1i$, we get $p_1 = 0.936 +0.815i$, $p_2 = 0.188 + 0.601i$, $p_3 = -0.296 +0.604i$, which converges after 6 iterations to $p_6 = -0.325 + 0.785i$

Even with poor starting points we often get convergence.

COMPLEX NEWTON-RAPHSON FOR POLYNOMIALS (file: complxnr_poly.f)

For $f(x) = x^n + a_{n-1}x^{n-1} + a_{n-2}x^{n-2} + \ldots + a_1x + a_0 = 0$ if the a’s are real, then all of the roots are either real or complex conjugates $A \pm Bi$.

Consider $f(x) = x^4 + x^3 – 3x^2 – 17x -30 = 0$

Starting with the initial value of $p_0 = 1 + 2i$, we get $p_1 = 0.173 + 1.140i$, $p_2 = -1.138 + 0.801i$, which converges after 9 iterations to $p_9 = -2.000 + 0.000i$, a real root.

The polynomial is now be reduced to $x^3 - x^2 - x - 15$. Starting with the same p_0 we get $p_1 = -0.077 + 0.615i$, $p_2 = -4.877 + 3.938i$, …, $p_8 = -1.000 + 2.000i$. With this complex root there is a conjugate of -1.000 – 2.000i. Now factoring the $(x+1-2i)(x+1+2i) = x^2 + 2x + 5$ out of the cubic we are left with the linear equation $x – 3 = 0$ giving the final root of 3. We have found 4 roots, 2 real, 2 complex conjugates: -2, 3, $-1\pm2i$.

BAIRSTOW’S ALGORITHM FOR POLYNOMIALS (file: Bairstownew.f)

$$f(x) = x^n + a_{n-1}x^{n-1} + a_{n-2}x^{n-2} + \ldots + a_1x + a_0 = (x^2 + px + q)(x^{n-2} + b_{n-3}x^{n-3} + \ldots + b_0) + Rx + Q$$

If we find the correct factor p and q, then there will be no remainder $Rx + Q$. R and Q will be 0. Bairstow’s algorithm starts with an initial $x^2 + p_0x + q_0$ and successive values of p and q are found converging quadratically to a factor with no remainder. Once a quadratic factor is found, the polynomial is decreased by two degrees until there is only a linear or quadratic polynomial remaining. The quadratic factor can produce two real roots or two complex conjugates.

Consider $f(x) = x^4 - 2x^3 + 26x^2 + 38x + 145 = 0$

Starting with the factor $x^2 + x + 1$, the next estimate is $x^2 + 1.960x + 5.191$, then the factor $x^2 + 2.000x + 4.998$, and then $x^2 + 2x + 5$. This gives the roots $-1 \pm 2i$ and the reduced polynomial is a quadratic with roots $2 \pm 5i$.

Program Listing - General Complex Newton-Raphson

```
      Complex Y,YP,P0,P1,F,FP,Z
      F(Z)=CEXP(Z)+Z*Z
      FP(Z)=CEXP(Z)+2*Z
      IT=0
      NMAX=20
      PRINT *,'Initial Complex Estimate.  Give as a,b where p0=a+bi'
      READ *,a,b
      P0=a*(1,0)+b*(0,1)
10    P1=P0-F(P0)/FP(P0)
      IF (CABS(P1-P0).LT.1.0E-6) THEN
        PRINT *, 'Converged'
        PRINT *, 'Root: ',REAL(P1),' + ', AIMAG(P1), 'i'
      ELSE
        P0=P1
        IT=IT+1
        PRINT *, 'P(',IT,')=',REAL(P0),' + ',AIMAG(P0),'i'
        IF (IT.LE.NMAX) THEN
          GOTO 10
        ENDIF
      ENDIF
      END
```

Program Output

```
 Initial Complex Estimate.  Give as a,b where p0=a+bi
2 1
 P( 1)=  0.935749233 +   0.815822303i
 P( 2)=  0.188221291 +   0.601407528i
 P( 3)= -0.295907617 +   0.604025722i
 P( 4)= -0.338343531 +   0.808917582i
 P( 5)= -0.325594038 +   0.785442948i
 P( 6)= -0.325199366 +   0.785257101i
 Converged
 Root:   -0.325199306 +   0.785257161i
```

Complex Newton-Raphson for Polynomials **COMPLXNR_POLY**

```
      DIMENSION A(0:20),B(0:20),RT(20,2)
      COMPLEX Y,YP,P0,P1,P(0:50)
      NMAX=20
      IC=1
      PRINT *,' N='
      READ *,N
      A(N)=1
      ND=N
      DO 20 I=N-1,0,-1
        WRITE(6,10)I
10      FORMAT(' Coefficient of X**',I3)
        READ *, A(I)
20    CONTINUE
22    PRINT *, 'Initial Complex Estimate. Give as a,b where p0=a+bi:'
      READ *, R,S
      P0=R*(1,0)+S*(0,1)
      P(0)=P0
      IT=0
25    Y=P0
      DO 30 I=N-1,1,-1
        Y=(Y+A(I))*P0
30    CONTINUE
      Y=Y+A(0)
      YP=N*P0+A(N-1)*(N-1)
      IF (N.GT.2) THEN
        DO 40 I=N-2,1,-1
          YP=YP*P0+A(I)*I
40      CONTINUE
      ENDIF
      P1=P0-Y/YP
      IF (CABS(P1-P0).LT.1.0E-6) THEN
        NMAX=IT
        GOTO 45
      ENDIF
      P0=P1
      IT=IT+1
      P(IT)=P1
      IF (IT.LT.NMAX) GOTO 25
45    DO 50 I=0,NMAX
        PRINT *, I,P(I)
50    CONTINUE
      RP=REAL(P1)
      XP=IMAG(P1)
      IF (ABS(XP).LE.1E-7) THEN
        PRINT *, ' ROOT=',RP
        RT(IC,1)=RP
        RT(IC,2)=0
        IC=IC+1
        IF (N.GE.2) THEN
          B(N-1)=1
          DO 54 K=N-2,0,-1
            B(K)=A(K+1)+RP*B(K+1)
54        CONTINUE
          N=N-1
          DO 55 I=0,N
            A(I)=B(I)
55        CONTINUE
        ENDIF
      ELSE
        PRINT *, ' ROOT=',RP,'   +/-',XP,'i'
```

Sample Input/Output: $x^4 + x^3 - 3x^2 - 17x - 30 = 0$

```
 N=
4
 Coefficient of X**  3
1
 Coefficient of X**  2
-3
 Coefficient of X**  1
-17
 Coefficient of X**  0
-30
 Initial Complex Estimate. Give as a,b where p0=a+bi:
1 2
 0 (1.,2.)
 1 (0.172602743,1.13972604)
 2 (-1.13765359,0.801370144)
 3 (-3.87670636,-0.740367651)
 4 (-2.99458337,-0.541453242)
 5 (-2.36668587,-0.35130161)
 6 (-2.02656984,-0.134431094)
 7 (-1.98977101,-0.00468340982)
 8 (-2.00004983,5.77036335E-005)
 9 (-2.,3.44612339E-009)
  ROOT= -2.
 New coefficients of reduced polynomial:
  A( 0)= -15.
  A( 1)= -1.
  A( 2)= -1.
  A( 3)=  1.
 Initial Complex Estimate. Give as a,b where p0=a+bi:
1 2
 0 (1.,2.)
 1 (-0.0769230798,0.615384638)
 2 (-4.87692308,3.93846107)
 3 (-3.14442325,2.71432352)
 4 (-1.97877085,2.0255475)
 5 (-1.23146391,1.82753253)
 6 (-0.980555058,1.97202897)
 7 (-1.00030613,2.00036502)
 8 (-1.,2.)
  ROOT= -1.    +/-  2.i
 New coefficients of reduced polynomial:
  A( 0)= -3.
  A( 1)=  1.
 Root:  3.

  ****** All Roots *******
  ------------------------
      Real        Imaginary
    -2.000000     0.000000
    -1.000000     2.000000
    -1.000000    -2.000000
     3.000000     0.000000
```

Roots: -2, $-1 \pm 2i$, 3 (note: initial guess $1 + 2i$)

```
        RT(IC,1)=RP
        RT(IC,2)=XP
        RT(IC+1,1)=RP
        RT(IC+1,2)=-XP
        IC=IC+2
        C1=-2*RP
        C0=RP*RP+XP*XP
        B(N-2)=1
        B(N-3)=A(N-1)-C1*B(N-2)
        N=N-2
        IF (N.GE.2) THEN
          DO 60 K=N-2,0,-1
            B(K)=A(K+2)-C1*B(K+1)-C0*B(K+2)
60        CONTINUE
        ENDIF
        DO 70 I=0,N
          A(I)=B(I)
70      CONTINUE
      ENDIF
      PRINT *,'New coefficients of reduced polynomial:'
      DO 80 I=0,N
        PRINT *,' A(',I,')=',A(I)
80    CONTINUE
      IF (N.GT.2) THEN
        NMAX=20
        GOTO 22
      ELSE
        IF (N.EQ.2) THEN
          Z=A(1)*A(1)-4*A(0)
          IF (Z.GE.0) THEN
            PRINT *,'Roots:',(-A(1)+SQRT(Z))/2,',',(-A(1)-SQRT(Z))/2
            RT(IC,1)=(-A(1)+SQRT(Z))/2
            RT(IC,2)=0
            RT(IC+1,1)=(-A(1)-SQRT(Z))/2
            RT(IC+1,2)=0
          ELSE
            PRINT *,'Roots:',-A(1)/2,' +/- ',SQRT(-Z)/2,'i'
            RT(IC,1)=-A(1)/2
            RT(IC,2)=SQRT(-Z)/2
            RT(IC+1,1)=-A(1)/2
            RT(IC+1,2)=-SQRT(-Z)/2
          ENDIF
          IC=IC+2
        ENDIF
        IF (N.EQ.1) THEN
          PRINT *, 'Root:',-A(0)
          RT(IC,1)=-A(0)
          RT(IC,2)=0
        ENDIF
      ENDIF
      WRITE(6,88)
88    FORMAT(/'  ****** All Roots *******'/'  .......................'/
     1  '      Real       Imaginary')
      WRITE(6,90)(RT(I,1),RT(I,2),I=1,ND)
90    FORMAT(2F13.6)
      END
```

```
C*********************************************************************
C                                                                    *
C       BAIRSTOW'S ALGORITHM - Written by Prof. Richard Goldstein    *
C                                                                    *
C*********************************************************************
C
C      FIND ALL OF THE ROOTS OF
C      F(X) = X**N + A(N-1)*X**(N-1) + ... + A(1)*X + A(0)
C      USING QUADRATIC FACTORS X**2 + P*X + Q
C
C      INPUT:    DEGREE N; COEFFICIENTS A(N-1), ..., A(1), A(0); EPS;
C               P0, Q0
C
C      OUTPUT:   ALL ROOTS
C
C      DEFINE ARRAYS
       DIMENSION A(0:50), B(0:50), C(0:50), RT(50,2)
       CHARACTER QQ*1, NAME1*16
C      GET INPUT
       PRINT *, ' Input degree of polynomial (integer) N ='
       READ *, N
       DO 10 I=N-1,0,-1
         PRINT *, ' Input coefficient of X** ',I
         READ *, A(I)
10     CONTINUE
       NC=0
       NRT=1
       PRINT *, ' Tolerance (eps) ='
       READ *, EPS
15     PRINT *, ' Maximum number of iterations ='
       READ *, NMAX
       WRITE(6,*) 'Select output destinations: '
       WRITE(6,*) '1. Screen '
       WRITE(6,*) '2. Text file '
       WRITE(6,*) 'Enter 1 or 2 '
       WRITE(6,*) ' '
       READ(5,*) FLAG
       IF (FLAG .EQ. 2) THEN
          WRITE(6,*) 'Input the file name in the form - '
          WRITE(6,*) 'drive:name.ext'
          WRITE(6,*) 'as example:  a:output.txt '
          WRITE(6,*) ' '
          READ(5,16) NAME1
16        FORMAT(A16)
          IOUT = 3
          OPEN(UNIT=IOUT,FILE=NAME1)
       ELSE
          IOUT = 6
       ENDIF
       IF (NC.GT.0) THEN
         PRINT *, 'Continue with current P, Q (Y/N)?'
         READ(5,18)QQ
18       FORMAT(A1)
         IF ((QQ.EQ.'Y').OR.(QQ.EQ.'y')) THEN
           NC=0
           GOTO 20
         ENDIF
       ENDIF
C      INITIALIZE QUADRATIC FACTOR
       PRINT *, ' Initial P0, Q0 ='
       READ *, P0, Q0
       P=P0
       Q=Q0
```

```
      NC=0
C     FIND REDUCED POLYNOMIALS B AND C
20    B(N-1)=A(N-1)-P
      B(N-2)=A(N-2)-P*B(N-1)-Q
      DO 30 I=3,N
        B(N-I)=A(N-I)-P*B(N-I+1)-Q*B(N-I+2)
30    CONTINUE
      C(N-1)=B(N-1)-P
      C(N-2)=B(N-2)-P*C(N-1)-Q
      DO 40 I=3,N
        C(N-I)=B(N-I)-P*C(N-I+1)-Q*C(N-I+2)
40    CONTINUE
      R=1
      IF (N.GT.3) THEN
        R=C(3)
      ENDIF
      D=C(2)*C(2)-R*(C(1)-B(1))
      IF ((ABS(D).LT.1.0E-7).OR.(ABS(D).GT.1.0E20)) THEN
        PRINT *,' Determinant is out of bounds ...'
        GOTO 15
      ENDIF
      P1=(B(1)*C(2)-B(0)*R)/D
      Q1=(B(0)*C(2)-B(1)*(C(1)-B(1)))/D
      AERR=ABS(P1)+ABS(Q1)
      IF (AERR.GT.EPS) THEN
        WRITE(IOUT,100)P,Q,AERR
100     FORMAT(' P = ',F12.6,'  Q = ',F12.6,'  Total Error = ',E15.5)
        P=P+P1
        Q=Q+Q1
        NC=NC+1
        IF (NC.GE.NMAX) GOTO 15
        GOTO 20
      ENDIF
50    WRITE (IOUT,100)P,Q,AERR
C     SOLVE QUADRATIC EQUATION FOR ROOTS
      Z=P*P-4*Q
      IF (Z.GE.0.0) THEN
        Z=SQRT(Z)
        R1=(-P+Z)*0.5
        R2=(-P-Z)*0.5
        RT(NRT,1)=R1
        RT(NRT,2)=0
        RT(NRT+1,1)=R2
        RT(NRT+1,2)=0
        NRT=NRT+2
        WRITE(IOUT,200)R1,R2
200     FORMAT(/,' Real Roots: ',2F15.7)
      ELSE
        Z=SQRT(-Z)
        R1=-0.5*P
        R2=0.5*Z
        RT(NRT,1)=R1
        RT(NRT,2)=R2
        RT(NRT+1,1)=R1
        RT(NRT+1,2)=-R2
        NRT=NRT+2
        WRITE(IOUT,300)R1,R2
300     FORMAT(/,' Complex Roots: ',F15.7,' +/- ',F15.7,' i')
      ENDIF
      IF (N.GT.4) THEN
        DO 60 I=0,N-3
          A(I)=B(I+2)
60      CONTINUE
```

```
          N=N-2
          P=P0
          Q=Q0
          GOTO 15
        ELSE IF (N.EQ.4) THEN
          P=B(3)
          Q=B(2)
          N=2
          GOTO 50
        ELSE IF (N.EQ.3) THEN
          R=-B(2)
          RT(NRT,1)=R
          RT(NRT,2)=0
          NRT=NRT+1
          WRITE (IOUT,400)R
 400      FORMAT(/,' Real Root: ',F15.7)
        ENDIF
        WRITE(IOUT,420)
 420    FORMAT(/'Roots:'/'         Real Part    Imaginary Part')
        DO 460 I=1,NRT-1
          WRITE(IOUT,440)RT(I,1),RT(I,2)
 440      FORMAT(2E18.7)
 460    CONTINUE
        CLOSE(UNIT=IOUT)
        PRINT *, 'Execution Completed'
        END
```

OUTPUT - BAIRSTOW'S METHOD

```
  Input degree of polynomial (integer) N =
4
  Input coefficient of X**  3
-2
  Input coefficient of X**  2
26
  Input coefficient of X**  1
38
  Input coefficient of X**  0
145
  Tolerance (eps) =
1e-6
  Maximum number of iterations =
20
 Select output destinations:
 1. Screen
 2. Text file
 Enter 1 or 2

1
  Initial P0, Q0 =
1 1
 P =     1.000000  Q =     1.000000  Total Error =     0.51512E+01
 P =     1.960141  Q =     5.191090  Total Error =     0.23460E+00
 P =     2.000214  Q =     4.996568  Total Error =     0.36460E-02
 P =     2.000000  Q =     5.000000  Total Error =     0.56727E-06

 Complex Roots:      -1.0000000 +/-       1.9999999 i
 P =    -4.000000  Q =    29.000000  Total Error =     0.56727E-06

 Complex Roots:       2.0000000 +/-       5.0000000 i

Roots:
         Real Part    Imaginary Part
    -0.1000000E+01     0.2000000E+01
    -0.1000000E+01    -0.2000000E+01
     0.2000000E+01     0.5000000E+01
     0.2000000E+01    -0.5000000E+01
```